Shear Wall Analysis

One useful application of the computational stiffness method is shear wall analysis. The objective in this type of analysis is to determine the forces on individual shear walls due to a global force on a lateral force resisting system consisting of columns and shear walls. The analysis procedure follows that of the stiffness method. The figure below shows the plan view of a generic building and one of the supporting shear walls. The relationship between the DOFs of the floor, $\mathbf{u}_{\text{floor}}$, and the DOFs of the shear wall, $\mathbf{u}_{\text{wall}}$, is sought for the analysis. The DOFs of the floor are arbitrarily selected to originate in the lower left corner. Setting the DOFs equal to one, one at a time, establishes the columns of the transformation matrix:

$$\mathbf{T}_{\text{wf}} = \begin{pmatrix} 1 & 0 & b \\ 0 & 1 & -a \\ 0 & 0 & 1 \end{pmatrix}$$



The load vector for the shown floor is:

$$\mathrm{F} = \begin{pmatrix} 0 \\ F \\ -c\,F \end{pmatrix}$$

Problem Definition

The figure below shows the plan view of a rectangular floor of a building. The gray-shaded components, i.e., four columns and two shear walls, form the lateral force resisting system, resisting the force F. The thickness of both shear walls is denoted t, and the columns measure t by t. The storey height is denoted by H. The objective is to determine the force on each column and shear wall caused by F. Geometric stiffness should be ignored, while shear deformation should be included for the two shear walls, as applicable, but not for the columns. Torsion should be included for all components, assuming the cross-sections are free to warp at all locations.



The problem is solved using matrix structural analysis, with degrees of freedom shown in the figure below. The floor is assumed to be rigid in-plane; hence, it has three degrees of freedom as shown. Similarly, each column and shear wall has three degrees of freedom, also shown in the figure below. Following that numbering, in the calculations below, the numbers and 1 and 2 are used to denote the 1 and 2 directions followed by the degrees of freedom. That means the 1-axis is the conventional y-axis. The 2-axis is the conventional z-axis.



Input [kN, m]

```
E = 60 × 10^6;
ν = 0.2;
t = 0.2;
L1 = 20;
L2 = 10;
L3 = 10;
L4 = 2;
L5 = 14;
L6 = 2;
H = 4.5;
F = 3120;
```

Shear modulus:

```
G = E / (2 × (1 + ν));
```

Wall A

Cross-section area:

```
A = 2 L4 t + L6 t
```

which yields: 1.2

Moment of inertia about the 1-axis:

$$I1 = \frac{t\,L6^3}{12} + 2\,L4\,t\left(\frac{L6}{2}\right)^2 \;//\; N$$

which yields: 0.933333

Centroid location along the 1-axis relative to the centre-line of the web:

$$y0 = \frac{(2\,L4\,t)\,\frac{L4}{2}}{A} \;//\; N$$

which yields: 0.666667

Moment of inertia about the 2-axis:

$$I2 = \left(\frac{L6\,t^3}{12} + L6\,t\,y0^2\right) + 2\left(\frac{t\,L4^3}{12} + L4\,t\left(\frac{L4}{2} - y0\right)^2\right)$$

which yields: 0.534667

Maximum shear flow in the flanges due to a unit shear force in the 2-direction (needed for shear centre calculation):

$$qB = \frac{1}{I1}\left(L4\,t\,\frac{L6}{2}\right)$$

which yields: 0.428571

Corresponding maximum shear flow at the neutral axis of the web:

$$qNA = qB + \frac{1}{I1}\left(t\,\frac{L6}{2}\,\frac{L6}{4}\right)$$

which yields: 0.535714

Resultant force of shear flow in flanges:

$$qForceFlange = \frac{1}{2}\,qB\,L4$$

which yields: 0.428571

Shear centre location relative to the centre line of the web:

```
equation = qForceFlange L6 == 1 distance;
solution = Solve[equation, distance];
ySC = (distance /. solution)[[1]]
```

which yields: 0.857143

Shear area in the 1-direction:

$$Av1 = 2\,L4\,t$$

which yields: 0.8

Alpha-coefficient for shear deformation in the 1-direction:

$$\alpha 1 = \frac{12\, E\, I2}{G\, Av1\, H^2}$$

which yields: 0.950519

Shear area in the 2-direction:

$$Av2 = L6\, t$$

which yields: 0.4

Alpha-coefficient for shear deformation in the 2-direction:

$$\alpha 2 = \frac{12\, E\, I1}{G\, Av2\, H^2}$$

which yields: 3.31852

St. Venant torsion coefficient:

$$J = \frac{1}{3} L6\, t^3 + 2 \times \frac{1}{3} L4\, t^3$$

which yields: 0.016

Stiffness matrix:

$$KA = \left\{\left\{\frac{12\, E\, I2}{(1+\alpha 1)\, H^3}, 0, 0\right\}, \left\{0, \frac{12\, E\, I1}{(1+\alpha 2)\, H^3}, 0\right\}, \left\{0, 0, \frac{G\, J}{H}\right\}\right\};$$

```
KA // MatrixForm
```

which yields:

$$\begin{pmatrix} 2.16585 \times 10^6 & 0 & 0 \\ 0 & 1.70764 \times 10^6 & 0 \\ 0 & 0 & 88\,888.9 \end{pmatrix}$$

Transformation matrix:

```
a = L1 - L5 - L4 - ySC;
b = L2/2;
TA = {{1, 0, b}, {0, 1, -a}, {0, 0, 1}};
TA // MatrixForm
```

which yields: $\begin{pmatrix} 1 & 0 & 5 \\ 0 & 1 & -3.14286 \\ 0 & 0 & 1 \end{pmatrix}$

Wall B

Moment of inertia about the 1-axis:

$$I1 = \frac{t\,L6^3}{12} \;//\; N$$

which yields: 0.133333

Moment of inertia about the 2-axis:

$$I2 = \frac{L6\,t^3}{12} \;//\; N$$

which yields: 0.00133333

Shear area in the 2-direction:

```
Av2 = L6 t
```

which yields: 0.4

Alpha-coefficient for shear deformation in the 1-direction:

$$\alpha 1 = 0;$$

Alpha-coefficient for shear deformation in the 2-direction:

$$\alpha 2 = \frac{12\,E\,I1}{G\,Av2\,H^2}$$

which yields: 0.474074

St. Venant torsion coefficient:

$$J = \frac{1}{3} L6\, t^3$$

which yields: 0.00533333

Stiffness matrix:

$$KB = \left\{\left\{\frac{12\, E\, I2}{(1+\alpha 1)\, H^3}, 0, 0\right\}, \left\{0, \frac{12\, E\, I1}{(1+\alpha 2)\, H^3}, 0\right\}, \left\{0, 0, \frac{G\, J}{H}\right\}\right\};$$

```
KB // MatrixForm
```

which yields:

$$\begin{pmatrix} 10\,535. & 0 & 0 \\ 0 & 714\,685. & 0 \\ 0 & 0 & 29\,629.6 \end{pmatrix}$$

Transformation matrix:

$$a = L1;$$

$$b = \frac{L2}{2};$$

```
TB = {{1, 0, b}, {0, 1, -a}, {0, 0, 1}};
TB // MatrixForm
```

which yields:

$$\begin{pmatrix} 1 & 0 & 5 \\ 0 & 1 & -20 \\ 0 & 0 & 1 \end{pmatrix}$$

Columns

Moments of inertia:

$$I1 = \frac{t^4}{12};$$

$$I2 = I1$$

which yields: 0.000133333

Stiffness matrix:

```
Kcolumns = {{12 E I2/H^3, 0, 0}, {0, 12 E I1/H^3, 0}, {0, 0, 0}};
Kcolumns // MatrixForm
```

which yields: $\begin{pmatrix} 1053.5 & 0 & 0 \\ 0 & 1053.5 & 0 \\ 0 & 0 & 0 \end{pmatrix}$

Transformation matrices:

```
TC = {{1, 0, L2}, {0, 1, 0}, {0, 0, 1}};
TD = {{1, 0, L2}, {0, 1, -L1}, {0, 0, 1}};
TE = {{1, 0, 0}, {0, 1, -L1}, {0, 0, 1}};
TF = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
```

Stiffness matrix for the entire floor

Stiffness matrix:

```
Kfloor = TA^T.KA.TA + TB^T.KB.TB + TC^T.Kcolumns.TC + TD^T.Kcolumns.TD +
   TE^T.Kcolumns.TE + TF^T.Kcolumns.TF;
Kfloor // MatrixForm
```

which yields: $\begin{pmatrix} 2.1806\times10^{6} & 0. & 1.0903\times10^{7} \\ 0. & 2.42654\times10^{6} & -1.97027\times10^{7} \\ 1.0903\times10^{7} & -1.97027\times10^{7} & 3.58323\times10^{8} \end{pmatrix}$

Load vector:

```
Ffloor = {0, F, -F L3};
Ffloor // MatrixForm
```

which yields: $\begin{pmatrix} 0 \\ 3120 \\ -31200 \end{pmatrix}$

Solve for the displacements of the floor:

```
uFloor = LinearSolve[Kfloor, Ffloor];
uFloor // MatrixForm
```

which yields: $\begin{pmatrix} 0.000203946 \\ 0.000954587 \\ -0.0000407891 \end{pmatrix}$

```
Inverse[Kfloor].Ffloor
```

which yields: {0.000203946, 0.000954587, -0.0000407891}

Recover forces in all shear walls and columns

```
FA = KA.TA.uFloor
```

which yields: {0., 1849., -3.6257}

```
FB = KB.TB.uFloor
```

which yields: $\{-4.44089 \times 10^{-16}, 1265.26, -1.20857\}$

```
FC = Kcolumns.TC.uFloor
```

which yields: {-0.214856, 1.00565, 0.}

```
FD = Kcolumns.TD.uFloor
```

which yields: {-0.214856, 1.86508, 0.}

```
FE = Kcolumns.TE.uFloor
```

which yields: {0.214856, 1.86508, 0.}

```
FF = Kcolumns.TF.uFloor
```

which yields: {0.214856, 1.00565, 0.}

Equilibrium check

Sum of forces in the 1-direction (should add up to zero):

```
FA[[1]] + FB[[1]] + FC[[1]] + FD[[1]] + FE[[1]] + FF[[1]]
```

which yields: -5.55112×10^{-16}

Sum of forces in the 2-direction (should add up to *F*):

```
FA[[2]] + FB[[2]] + FC[[2]] + FD[[2]] + FE[[2]] + FF[[2]]
```

which yields: 3120.